Create a Language of The Month answer scraper When I was posting answers for the Husk Language of The Month thread, I realized it took more time to enter them in the thread than it took to actually write the answers. So, why not automate it?
Write a program/function which inputs a language name, and displays all answers in that language during the current month so far)
Sandbox
Introduction
You are to find all answers in Code Golf Stack Exchange in the current month so far, whose first non-empty line matches the following:
#&lt;whitespace&gt;[&lt;language name&gt;(case insensitive match)]&lt;anything else&gt;

For example, the current month's answers should have been posted within the time frame of Nov 1 till the current date(Nov 6, as of this question's sandbox post).
for each answer, you must find the following:

Question name
Answer URL
Answer Poster's Name
Answer Poster's Profile URL

And format them in the following manner for Markdown:
- [&lt;Question Name&gt;](Answer URL) by [&lt;Answer Poster's Name&gt;]

Once you are done listing the answers, to keep the post short, you must key in the usernames to their respective profile URLs as follows:
- [&lt;Answer Poster's Name&gt;]:&lt;Answer Poster's Profile URL&gt;

This must be added at the end of the output.
Output
The final result should look like the following:
- [Size up a number](http://example1.com) by [Razetime]
- [My smartphone's Phonebook](http://example2.com) by [Bubbler]
- [1,2,Fizz,4,Buzz](http://example3.com) by [Lyxal]
- [Upgoat or Downgoat?](http://example3.com) by [LegionMammal978]
- .....

[Razetime]:http://exampleA.com
[Bubbler]:http://exampleB.com
[Lyxal]:http://exampleC.com
[LegionMammal978]:http://exampleD.com
...

Scoring
This is code-golf. Shortest answer in each language wins.
05AB1E, 283 282 bytes
žgŽ7?-©ƵÇ+400÷®&lt;т÷(®69-4÷®70-365*•ΘÏF•ºS₂+žf&lt;£`žgт‰0Kθ4ÖOŽª+·*’¸¸.Â‚‹º.ŒŒ/–«?€¼=ƒËŠˆ&amp;€š‚‚=ÿ’žYì.w„†ä_¡¦ε',¡нþ}Ùε’ƒËŠˆ.Â‚‹º.ŒŒ/’žYì©’ÿ…é/ÿ’.w……&lt;h1¡¦ε„&gt;&lt;S¡3èy’=&quot;/a/’¡1è'&quot;¡н®&quot;ÿa/ÿ&quot;y'†§™’ÜÇ.µÔ/ÿ’¡1è…&gt;&lt;&quot;S¡64Sè`®¨ì)}ćнδª}€`ʒнlIlQ}ʒ1è'/¡4èXså]DεƵ=Sè`“- [ÿ](ÿ)€‹ [ÿ]“}»sε32Sè`&quot;[ÿ]:ÿ&quot;}Ù»¶2×ý

Ugh..
No TIO of the full program, because .w is disabled.
Explanation:

Step 1: Get the current year and month, and convert it to a Unix timestamp (using this formula):
žg                     # Push the current year
  Ž7?-                 # Subtract 1900
      ©                # Store it in variable `®` (without popping)
       ƵÇ+             # Add 299
          400÷         # Integer-divide by 400
      ®                # Push year-1900 again from variable `®`
       &lt;               # Subtract 1
        т÷             # Integer-divide by 100
          (            # Negate it
      ®                # Push year-1900 again
       69-             # Subtract 69
          D            # Duplicate it
           4÷          # Integer-divide the copy by 4
          s            # Swap so year-1969 is at the top again
           &lt;           # Subtract 1 to make it year-1970
            365*       # Multiply by 365
•ΘÏF•                  # Push compressed integer 5254545
     º                 # Mirror it to 52545455454525
      S                # Convert it to a list of digits
       ₂+              # Add 26 to each: [31,28,31,30,31,30,31,31,30,31,30,31,28,31]
         žf            # Push the current month
           &lt;           # Subtract it by 1
            £          # Leave that many trailing items in the list
             `         # Pop and push them all separated to the stack
žg                     # Push the current year again
  т‰                   # Divmod by 100: [year//100,year%100]
    0K                 # Remove 0s from this pair
      θ                # Leave the last item
       4Ö              # Check if it's divisible by 4
                       # (this results in 1 for leap years; 0 for non-leap years)
O                      # Sum all values on the stack
 Žª+                   # Push compressed integer 43200
    ·                  # Double it to 86400
     *                 # Multiply the sum by this 86400

Try just this first step online.
Step 2: Use this timestamp to create the stackexchange-API URL, and get the content of this url (which contains the JSON of all answers created this month):
’¸¸.Â‚‹º.ŒŒ/–«?€¼=ƒËŠˆ&amp;€š‚‚=ÿ’
                       # Push dictionary string &quot;api.stackexchange.com/answers?site=codegolf&amp;fromdate=ÿ&quot;,
                       # where the `ÿ` is automatically replaced with the timestamp
 žYì                   # Prepend constant &quot;https://&quot;
    .w                 # Browse to this URL, and read its contents as string

Try the first two steps online (without .w).
Step 3: Extract all unique answer-id and question-ids from this JSON:
D                      # Duplicate the JSON
 „†ä_¡                 # Split it on dictionary string &quot;question_&quot;
      s                # Swap so the JSON is at the top again
       „Ž»_¡           # Split it on dictionary string &quot;answer_&quot;
            ‚          # Pair those two lists together
ε                      # Map over the pairs:
 ¦                     #  Remove the first item
  ε                    #  Map over each inner string:
   ',¡                '#   Split it on &quot;,&quot;
      н                #   Pop and leave just the first item
       þ               #   Pop and leave just the digits in this string
  }Ù                   #  After the inner map: uniquify the resulting ids
}                      # Close the outer map

Try just this third step with the JSON as input.
Step 4: Convert each question-id to an URL, and get its contents (which is the entire HTML page of a question):
`                      # Pop and push both lists separated to the stack
 U                     # Pop the list of answer-ids and store it in variable `X`
ε                      # Map over the question-ids:
 ’ƒËŠˆ.Â‚‹º.ŒŒ/’       #  Push dictionary string &quot;codegolf.stackexchange.com/&quot;
   žYì                 #  Prepend constant &quot;https://&quot;
      ©                #  Store this string in variable `®` (without popping)
       ’ÿ…é/ÿ’         #  Push dictionary string &quot;ÿ/questions/ÿ&quot;,
                       #  where the first `ÿ` is automatically replaced with &quot;https://codegolf.stackexchange.com/&quot;
                       #  and the second `ÿ` is automatically replaced with the current question-id
              .w       #  Browse to this URL, and read its contents as string

Try steps 3 and 4 with JSON input (without .w).
Step 5: For each HTML-content of a question, get a list of its [language_name, answer_url, user_name, user_url, question_name] (this may contain some garbage in the result - i.e. the trailing item, or if an answer contains multiple &lt;h1&gt;-refs, but we'll filter those out later):
…&lt;h1¡                  #  Split the HTML-content on &quot;&lt;h1&quot;
     ¦                 #  Remove the first item
      ε                #  Map over each string:
       „&gt;&lt;S¡           #   Split it on both &quot;&gt;&quot; and &quot;&lt;&quot;
            3è         #   Get the fourth item (which is the question name in the first iteration,
                       #   and language name in every other iteration)
       y               #   Push the string again
        ’=&quot;/a/’¡      &quot;#   Split it on &quot;=&quot;/a/&quot;
                1è     #   Get the second item (which is the trailing answer-id URL portion)
                  '&quot;¡ '#   Split that on '&quot;'
                     н #   Pop and leave the first item
        ®              #   Push &quot;https://codegolf.stackexchange.com/&quot; from variable `®`
         &quot;ÿa/ÿ&quot;        #   Push dictionary string &quot;ÿa/ÿ&quot;,
                       #   where the `ÿ` are filled again to create the answer-URL
       y               #   Push the string again
        '†§           '#   Push dictionary string &quot;person&quot;
           ™           #   Titlecase it to &quot;Person&quot;
            ’ÜÇ.µÔ/ÿ’  #   Push dictionary string &quot;schema.org/ÿ&quot;,
                       #   where the `ÿ` is replaced with &quot;Person&quot;
             ¡         #   Split the string on that
              1è       #   Get the second item
        …&gt;&lt;&quot;S¡        &quot;#   Split that on &quot;&gt;&quot;, &quot;&lt;&quot;, and '&quot;'
              64Sè     #   Get the seventh and fifth items
                       #   (which are the user name and trailing user URL portion)
                  `    #   Push those separated to the stack
                   ®   #   Push &quot;https://codegolf.stackexchange.com/&quot; again
                    ¨  #   Remove the trailing &quot;/&quot;
                     ì #   Prepend it in front to create the user-URL
       )               #   Wrap all four values into a list
      }ć               #  After the map: extract the first quadruple
        н              #  Pop and push its first item (the question name)
         δ             #  Map over each remaining quadruples
          ª            #   And append the question name as trailing item to each

Try just this fifth step with the HTML as input.
Step 6: Filter this list, so it only contains answers from the earlier list, and only the language that's equal to the input:
ʒ                      #  Filter this list of quintuples by:
 н                     #   Pop and leave its first item (the language name)
  l                    #   Convert it to lowercase
   Il                  #   Get the input in lowercase as well
     Q                 #   And check that both are equal
}ʒ                     #  After the filter: filter once again:
  1è                   #   Get the second item (the answer URL)
    '/¡               '#   Split it on &quot;/&quot;
       4è              #   Leave its fifth item (the answer-id)
         X             #   Push the list of answer-ids from variable `X`
          så           #   And check if the current answer-id is in it
]                      # Close both the filter and outer map

Try just the sixth step with the list of quintuples of the previous step as input.
Step 7: Convert it to the output in the specified format:
 €`                    # After the map over the questions, flatten it one level down
   D                   # Duplicate this list
    ε                  # Map over the quintuplets:
     Ƶ=                #  Push compressed integer 214
       Sè              #  Leave just the third, second, and fifth items
          `            #  Pop and push them separated to the stack
      “- [ÿ](ÿ)€‹ [ÿ]“ #  Push dictionary string &quot;- [ÿ](ÿ) by [ÿ]&quot;,
                       #  where the `ÿ` are filled again
    }»                 # After the map: join the strings by newlines
   sε                  # Swap, and map over the quintuplets again:
     32Sè              #  Leave just the fourth and third items
         `             #  Pop and push them separated to the stack
      &quot;[ÿ]:ÿ&quot;          #  Push string &quot;[ÿ]:ÿ&quot;,
                       #  where the `ÿ` are filled again
    }»                 # After the map: join the strings by newlines as well
      ¶2×ý             # Then join these two strings with &quot;\n\n&quot; as delimiter
                       # (after which the result is output implicitly)

Try just the sixth and seventh steps with the list of quintuplets of step five as input.
See this 05AB1E tip of mine (sections How to use the dictionary? and How to compress large integers?) to understand how the dictionary strings and compressed integers work.